

This lab explores the use of explicit Runge–Kutta methods of the form

$$\begin{aligned}\xi_1 &= y_n \\ \xi_2 &= y_n + ha_{21}f(t_n, \xi_1) \\ \xi_3 &= y_n + ha_{31}f(t_n, \xi_1) + ha_{32}f(t_n + c_2h, \xi_2) \\ &\vdots \\ \xi_\nu &= y_n + h \sum_{i=1}^{\nu-1} a_{\nu i}f(t_n + c_ih, \xi_i) \\ y_{n+1} &= y_n + h \sum_{j=1}^{\nu} b_jf(t_n + c_jh, \xi_j).\end{aligned}$$

Recall $\xi_i \approx y(t_n + c_ih)$ and $y_n \approx y(t_n)$ where $t_n = t_0 + hn$. Also,

$$c_1 = 0, \quad c_2 = a_{21}, \quad c_3 = a_{31} + a_{32}, \quad \dots, \quad c_\nu = \sum_{i=1}^{\nu-1} a_{\nu i}$$

and further that $\sum_{j=1}^{\nu} b_j = 1$.

Such a ν -stage method uses repeated nesting of the function f along with multiplication by h to obtain a higher-order approximation y_{n+1} from the lower-order terms represented by ξ_i .

While the form presented above is useful for analysis, it makes practical sense to identify common subexpressions so that f is evaluated as few times as possible. Setting $k_i = f(t_n + c_ih, \xi_i)$ thus leads to

$$\begin{aligned}k_1 &= f(t_n, y_n) \\ k_2 &= f(t_n + c_2h, y_n + ha_{21}k_1) \\ k_3 &= f(t_n + c_3h, y_n + h(a_{31}k_1 + a_{32}k_2)) \\ &\vdots \\ k_\nu &= f\left(t_n + c_\nu h, y_n + h \sum_{i=1}^{\nu-1} a_{\nu i}k_i\right) \\ y_{n+1} &= y_n + h \sum_{j=1}^{\nu} b_jk_j.\end{aligned}$$

In this lab we shall use the RK4 method with tableaux given by

$$\begin{array}{c|c} c & A \\ \hline & b \end{array} = \begin{array}{c|ccc} 0 & & & \\ \frac{1}{2} & \frac{1}{2} & & \\ \frac{1}{2} & 0 & \frac{1}{2} & \\ 1 & 0 & 0 & 1 \\ \hline & \frac{1}{6} & \frac{1}{3} & \frac{1}{3} & \frac{1}{6} \end{array}$$

to approximate solutions of a coupled system of three ordinary differential equations called the Lorenz system.

The Lorenz System

The Lorenz system is an autonomous three-dimensional ordinary differential equation of the form

$$\frac{dy}{dt} = f(y)$$

with a given initial condition $y(0) = y_0$ where $y(t)$ is a vector in \mathbf{R}^3 and

$$f(y) = \begin{bmatrix} -10y_1 + 10y_2 \\ 28y_1 - y_2 - y_1y_3 \\ y_1y_2 - (8/3)y_3 \end{bmatrix}.$$

Each person will have a different initial condition y_0 . Click on the following link to retrieve the values of your initial condition:

<https://fractal.math.unr.edu/~ejolson/467-23/y0/mky0.cgi>

Please do not use anyone else's initial condition for this lab.

To implement the RK4 method described above first write a subroutine to compute the function $f(y)$. In Julia this may be done with the code

```

3 function f(y)
4     r=[10*(y[2]-y[1]),
5       (28.0-y[3])*y[1]-y[2],
6       y[1]*y[2]- (8/3)*y[3]]
7     return r
8 end

```

The RK4 Timestep

Next, write a subroutine to make one RK4 timestep. Using the coefficients given in the tableaux to compute the k_i yields

```

10 function rk4(y,h)
11     k1=f(y)
12     k2=f(y+h*1/2*k1)
13     k3=f(y+h*1/2*k2)
14     k4=f(y+h*k3)
15     return y+h*(1/6*k1+1/3*k2+1/3*k3+1/6*k4)
16 end

```

Note Julia will use multiple dispatch to compile efficient versions of `rk4` for whatever length vectors appear as `y` in the arguments. The built-in vector notation then makes the code for solving systems of ordinary differential equations appear identical to the code for solving scalar equations.

Plotting the Solution

Our goal is to plot an approximation of the solution's trajectory in phase space for $t \in [0, T]$ where $T = 10$. This will yield a visualization of what has commonly been called the Lorenz butterfly in chaos theory and the study of nonlinear dynamics.

Consider approximating the solution using $N = 20480$ time steps of size $h = T/N$. The result is a sequence of 20480 vectors $y_n \in \mathbf{R}^3$. While the plotting system would likely handle 20480 points without trouble, it's not difficult to imagine lengthier calculations with even more points. Thus, it is reasonable to plot only a subsample of the total timesteps.

One way to do this is with two nested loops where the outer loop stores the points to be plotted while the inner loop advances a certain number of time steps to find the next suitable point for plotting. To make the code more straight forward, we place the inner loop in a separate subroutine `solve` that performs `n` steps of size `h`. In particular, we have

```

18 function solve(y0,h,n)
19     yn=copy(y0)
20     for j=1:n

```

```

21     yn=rk4(yn,h)
22     end
23     return yn
24 end

```

It's worth mentioning that line 19 copies the initial condition as `yn=copy(y0)` to prevent `y0` from getting overwritten. If instead line 19 appeared as `yn=y0` this would indicate `yn` is a pointer referencing `y0`. In that case any changes to `yn` would also change `y0`. The explicit `copy` avoids this pitfall.

Before writing the outer loop that repeatedly calls `solve` we need to decide how many `rk4` steps should be made between the points we plot. One doesn't want to plot so many points that the plotting library runs slowly or out of memory, nor does one want to plot so few points that the graph no longer appears like a smooth solution to a differential equation.

Given the apparent speed at which the dynamics in the Lorenz equations evolve and the fact that $h = 1/2048$, skipping every $m = 16$ timesteps between plotted points still yields a smooth curve. On the other hand, skipping every 16 timesteps reduces the number of points to plot from $N = 20480$ to $P = 1280$ which results in a graph that is efficient to render.

Code to calculate the relevant parameters used in the loops is

```

26 N=20480
27 T=10
28 h=T/N
29
30 m=16
31 P=N÷m

```

Note that line 31 includes the Unicode integer division operator `÷` rather than the usual `/` which would have resulted in a floating point value. This character can be entered in the Julia REPL by typing `\div` followed by the `<tab>` key. If you have difficulty typing `÷` into the editor try cut and paste from the REPL using the mouse.

One can initialize the arrays in which to store the points for plotting and write the outer loop as

```

33 X=zeros(P)

```


```

34 Y=zeros(P)
35 Z=zeros(P)
36
37 y0=[-2.20, -3.46, 16.67]
38 yj=copy(y0)
39 for j=1:P
40     global yj=solve(yj,h,m)
41     X[j]=yj[1]
42     Y[j]=yj[2]
43     Z[j]=yj[3]
44 end

```

The initial condition `y0` appearing on line 37 reflects the value when I click on the web link mentioned earlier. You will have to change this to your individualized initial condition. Line 40 includes a `global` declaration to resolve the ambiguity between `yj` in the global scope and the possibility of a local version of `yj` inside the scope of the loop.

Finally, to create a graph that looks similar to



plot the output using the `Plots` library.

```

46 using Plots
47 plot(X,Y,Z,label="$y0")
48 savefig("butterfly.pdf")

```

At this point you should have a file called `butterfly.pdf` stored in your working directory. If it looks *exactly* like the above figure, that may mean you forgot to change the initial condition.

Extra Credit

For extra credit refine the stepsize of your solution by making N larger (and h smaller) in order to approximate the solution $y(10)$ to three decimal places. Consider a sequence of smaller and smaller step sizes and show that the resulting approximations y_N appear to converge numerically. Can you find four decimal places? What about five?

Submitting Your Work

Two things should be uploaded for grading:

- A PDF file `lorenz.pdf` containing the code `lorenz.jl` used to generate the graph `butterfly.pdf`.
- The graph `butterfly.pdf` corresponding to your initial condition.

The files `butterfly.pdf` has already been created and should be in the `lab04` subdirectory. The only thing left is to convert `lorenz.jl` and its output into a PDF file for upload. In the lab the commands

```
$ j2pdf -o lorenz.pdf lorenz.jl
```

may be used to produce a file `lorenz.pdf` suitable for uploading. You may check your submission using `evince` to view the PDF files.

Before leaving don't forget to close the applications open on your desktop and logout. Exit the Julia REPL by typing `<ctrl>-d` and then `<ctrl>-d` again to close the terminal. The editor has a menu at the top. If using one of the lab computers, please reboot it into Microsoft Windows.